

## 8.2

## 要素削除によるデータ生成

(1)生成の考え方 図8-3に示すように、物体全体を覆うような長方形のデータを生成しておき、無効な部分を指定することで、有効な箇所だけを残すことでデータを生成する方法です。このほうが、データ自動生成が単純になるケースも数多くあります。




図8-3 要素削除によるデータ生成

(2)用意するシート 以下のような名前のシートを用意します。

- ①図 : 生成用のデータ指定および図を描画するシート。
- ②生成用節点 : 生成する前の節点データ。最終的に生成される節点データではなく、中間的なデータです。
- ③生成用要素 : 生成する前の要素データ。最終的に生成される要素データではなく、中間的なデータです。

【生成用データ】 「図」シートに以下のようなデータを指定します。なお、括弧内の B1, B2 等はセル位置です。

List 8-17 生成用データ図形表示

```

Sub drawElement(Eno, XX1, YY1, XX2, YY2, XX3, YY3, Visible)
  With Worksheets("図")
    Xmin = Val(.Cells(1, 4)) ' X=0 のときのシート内左座標
    Ymax = Val(.Cells(2, 4)) ' Y=0 のときのシート内下座標
    SC = Val(.Cells(3, 4)) ' スケール
  End With
  X1 = Xmin + XX1 * SC: Y1 = Ymax - YY1 * SC ' シート内座標に変換
                                           ' (上向きを正とする)
  X2 = Xmin + XX2 * SC: Y2 = Ymax - YY2 * SC
  X3 = Xmin + XX3 * SC: Y3 = Ymax - YY3 * SC
  ' 三角要素をフリーフォームとして生成
  With ActiveSheet.Shapes.BuildFreeform(msoEditingAuto, X1, Y1)
    .AddNodes msoSegmentLine, msoEditingAuto, X2, Y2
    .AddNodes msoSegmentLine, msoEditingAuto, X3, Y3
    .AddNodes msoSegmentLine, msoEditingAuto, X1, Y1
    .ConvertToShape.Select
  End With
  Selection.Name = "E_" & Eno ' 図形名の指定
  Selection.ShapeRange.Fill.Visible = Visible ' 図形を見えるモードに
  Selection.ShapeRange.Line.Visible = Visible ' する。
End Sub
Sub drawAllElements() ' 全要素の描画
  For i = 1 To NumberOfElement ' 要素数繰返し
    With Element(i)
      ii = .P1: X1 = Node(ii).X: Y1 = Node(ii).Y ' 3点座標設定
      ii = .P2: X2 = Node(ii).X: Y2 = Node(ii).Y
      ii = .P3: X3 = Node(ii).X: Y3 = Node(ii).Y
      drawElement i, X1, Y1, X2, Y2, X3, Y3, .Visible ' 1要素表示
    End With
  Next
End Sub

```

List 8-18 生成用データ設定の全体制御

```

Sub ボタン1_Click()
  clearData ' 生成用データのクリア
  generateInitialData ' 生成用データの初期設定
  NumberOfNode = getNumberOfSheetData("生成用節点", 1) - 1 ' 節点数
  NumberOfElement = getNumberOfSheetData("生成用要素", 1) - 1 ' 要素数
  ReDim Node(NumberOfNode), Element(NumberOfElement) ' 配列サイズの確保
  readGenNodeData ' 生成節点データを配列に読み込み
  readGenelementData ' 生成要素データを配列に読み込み
  Sheets("図").Select ' 描画用シート「図」を選択
  drawAllElements ' 全要素を描画
End Sub

```

#### (4) 有効・無効の切替え

初期設定された要素の図形を選択し、選択した図形を無効にしたり、有効にする処理です。なお、図形を選択した後、ボタンをクリックする前に「オブジェクトを選択」モードを解除して下さい。